Processing RapidEye Imagery

Megan M. Coffer
March 14th 2020
 

This documentation and the accompanying script were developed my Megan M. Coffer1,2
1 ORISE Fellow, U.S. Environmental Protection Agency, Office of Research and Development, Durham, NC
2 PhD Candidate, Center for Geospatial Analytics, North Carolina State University, Raleigh, NC

Permission is granted for an individual or institution to use this script provided that any publications developed using this script properly cite the manuscript accompanying this script:

Coffer, M. M., Schaeffer, B. A., Zimmerman, R. C, Hill, V., Li, J., Islam, K. A. & Whitman, P. J. (2020). Performance across WorldView-2 and RapidEye for reproducible seagrass mapping. Remote Sensing of Environment. DOI: 10.1016/j.rse.2020.112036.


This document accompanies the process_re_20200810.pro script with the objective of fully processing a RapidEye satellite scene from the raw imagery to at least an atmospherically corrected scene ready for classification, with the option to orthorectify and mosaic scenes into a single raster per overpass. 

RapidEye imagery is delivered from Planet Labs as zip files containing up to ten scenes per zip file (with each scene then containing multiple files depending on the spatial coverage of the scene). Given that a single zip files contains multiple scenes, filenames are difficult to discern from the zip file. Thus, RapidEye imagery were unzipped and renamed by hand, following the suggested file naming structure described below. This is different than the WorldView data processing script which unzips and names the data according to the zipped file filename. 
















Running this script – here, the basic workflow for opening and running the RapidEye script is described. 

1. Open the script in IDL. 
2. Change the rootDir variable in line 46 to the full path name up to an including the “RapidEye_Imagery” folder. This is further discussed below in Script components Step #3 Setting the root and working directories. 
3. Place the cursor anywhere inside the script editor window and click the “Compile” button along the top. The IDL Console window should print green text to indicate that the script was successfully compiled. If red text appears, there was an error compiling, typically because of a syntax error. 
4. Once the script has successfully compiled, type the name of the script followed by a comma and the name of the area of interest. The area of interest must be in single quotation marks and must match exactly the folder name that contains zip files for the area of interest (including capitalization, underscores, etc.). For example, after compiling, the following should be entered in the IDL console window: 

process_re_20200810, 'St_Joe'




























Script components – here, the code is described line-by-line and the necessary file structure is explained in detail. 

1. Defining the procedure and compiling IDL. An IDL script is known as a procedure, and it must be defined for the script to run properly. The start of this calls PRO which defines the procedure called process_re_20200314. After the procedure name, there is a comma and then the input variables. Here, only one input variable is required, called aoi. This variable is the area of interest to be processed in the script. When defining this variable, it is important to keep in mind that the variable name must follow exactly the AOI name of the folder that contains the input files (i.e. including capitalization, underscores, etc.).  

  PRO process_re_20200810, aoi

In addition to defining the procedure, IDL must be compiled. A COMPILE_OPT is a statement which is processed by the IDL compiler to change the behavior of the compiler. Specifying IDL2 sets two flags for the IDL compiler, DEFINT32 and STRICTARR. The DEFINT32 flag sets the default size of IDL integers to be 32-bits rather than 16-bits. The STRICTARR flag prevents IDL from using parentheses for array indexing. 


  ; Compile IDL
  COMPILE_OPT IDL2


2. Defining the Rayleigh Exponent. The Rayleigh exponent is one of the three components of atmospheric correction that can be adjusted. The Rayleigh exponent is used to compute the scattering factor and to compute the subtraction values for each band. A higher Rayleigh exponent will result in more correction applied to the imagery, whereas a lower Rayleigh exponent will result in less correction applied to the imagery. At several sites along the Florida coast and in the Chesapeake Bay, a Rayleigh exponent of 4.75 was found to be optimal for creating separability within the different spectral classes while maintaining positive reflectance values across all bands. However, this is a component that may need to be adjusted at other sites or for different time periods. 


  ; Define the Rayleigh exponent to use for atmospheric correction
  rayleighExp = 4.75


3. Setting the root and working directories. Unless the Rayleigh exponent requires adjusting, this should be the only line of code that needs to be manually changed before compiling and calling the script. 

  ; Define AOI and change working directory
  PRINT, 'Processing satellite imagery for ' + aoi 
  rootDir = 'E:\Estuary_Seagrass\RapidEye_Imagery\'
  workingDir = rootDir + aoi
[image: ]The rootDir line should specify the full path of the RapidEye_Imagery folder. Within that folder, a separate folder should exist for each AOI. Within each AOI folder (i.e. “Chesapeake_Aeronet” in the figure to the right), a folder should exist called “1_Zip_File.” These zip files should be named with the following naming convention:


AOI_SENSOR_YYYYMMDD_N.zip

AOI = area of interest (exact same name as AOI folder)
SENSOR = RE
YYYYMMDD = The date of image acquisition 
N = An index which is set to 1 except when there are multiple scenes collected per day



Additionally, the RapidEye_Imagery folder should contain a folder called InputData which contains a global DEM available with every ENVI/IDL installation (GMTED2010.jp2) and two subfolders: estuaries and extents. These subfolders contain an XML of the estuary boundary for the given AOI and an XML of the bounding box for the given AOI. Documentation for creating these input files are available at the DOI specified in the associated manuscript. The workingDir will then be set to the AOI folder within the rootDir. 


4. Suppressing automated messages. ENVI will automatically print many messages to the IDL Console. This line of code silences theses automated console messages. Thus, the only text printed to the IDL console as the procedure is run will be generated from within the script and provide updates on the progress of the script processing. 
  

  ; Suppress automated console messages
  !quiet = 1
















5. Listing satellite files. In order to properly include each file and associated filename in the processing, the next section of code finds all files and determines the unique satellite overpass dates for the given area of interest. The script first lists all folders containing '*_rawData' within the '\2_Raw_Data\' folder. The FILE_BASENAME function is then used to determine each filename. The unique number of characters in these basenames is returned and used to query the basename to determine the date of overpass and index (note, index is usually 1 unless there was more than one overpass per day). 

  ; List rawDataFiles for the given AOI
  rawDataFiles = FILE_SEARCH(workingDir + '\2_Raw_Data\', '*_rawData')
  ; Determine the basename of each file in rawDataFiles
  rawDataBasename = FILE_BASENAME(rawDataFiles)
  ; Find the unique number of characters in each element in rawDataBasename
  nCharacters = STRLEN(rawDataBasename)
  uniqueCharacters = nCharacters[UNIQ(nCharacters)]
  ; Determine the collection dates for each filename
  rawDataDates = STRMID(rawDataBasename,uniqueCharacters-18,10)
  ; Find unique dates in rawDataDates
  uniqueDates = rawDataDates[UNIQ(rawDataDates)]


6. Loop through 2_Raw_Data files to process each scene. Each satellite overpass is collected in a scene, which is then broken up into tiles. We first need to loop through the scenes listed in uniqueDates and list tiles contained within each scene. 


  ; Loop through each date in uniqueDates
  FOR i = 0, N_ELEMENTS(uniqueDates) - 1 DO BEGIN


7. Initiate ENVI and set temporary directory. First, ENVI is initiated from within IDL. This will open an ENVI window (throughout processing of this script, an ENVI window may open and display a progress bar, but there is no need to interact with the ENVI window). Additionally, IDL and ENVI create a lot of temporary files during processing. In an effort to prevent storage issues, a folder is created and designated for temporary files throughout processing. This folder can then be deleted after the script has run.


  ; Initiate ENVI session
  e = ENVI()


  ; Create and set a temporary folder
  prefs = e.Preferences
  tempFolder = FILEPATH('tempFolder\', ROOT_DIR = workingDir)
  FILE_MKDIR, tempFolder
  prefs['directories and files:temporary directory'].VALUE = tempFolder

8. Read in input files. The required input files were those listed in the InputData folder mentioned previously. The extent and estuaries XML files will be read in as ROI files, and the global DEM will be read in as a raster file. 


  ; Read in required input files
  ; extentRoi outlines the extent of the area of interest
  extentRoiPath = FILEPATH('InputData\extents\' + aoi + '_extent.xml', ROOT_DIR = rootDir)
  extentRoi = e.OpenRoi(extentRoiPath)
  ; estuariesRoi outlines the estuary for the area of interest 
  estuariesRoiPath = FILEPATH('InputData\estuaries\' + aoi + '_estuaries.xml', ROOT_DIR = rootDir)
  estuariesRoi = e.OpenRoi(estuariesRoiPath)
  ; demInput is a global DEM raster file 
  demFile = FILEPATH('InputData\GMTED2010.jp2', ROOT_DIR = rootDir)
  demInput = e.OpenRaster(demFile)


9. Subset to the ith rawData folder. Using the uniqueDates array created above, subset to the ith index and search for its associated files. A file exists for each tile within the scene. This will be indexed to loop through each tile. Print a progress update to the IDL console window.  


  ; Subset to the ith rawData file
  dateRawData = FILE_SEARCH(workingDir + '\2_Raw_Data\' + aoi + '_RE_' + uniqueDates[i] + '_rawData/' + '*_RE*')
  ; Print progress of the loop 
  PRINT, STRING(9B) + 'Processing imagery for date: ' + STRCOMPRESS(uniqueDates[i] + ' (File ' + STRING(i + 1) + ' of ' + STRING(N_ELEMENTS(uniqueDates)) + ')')



















10. Generate arrays to output scene information. This script will output information about each tile contained within the scene. The next section of code defines arrays to populate with varying information which will later be populated for each tile and saved as a CSV file. First, a rededgeList array is defined which will be populated with red edge values for each tile. This is important for atmospheric correction. 


  ; Define an array to populate with red edge values for each tile
  rededgeList = MAKE_ARRAY(N_ELEMENTS(dateRawData), 1, /FLOAT, VALUE = -9999)
  ; Define arrays to populate with image information
  tileFilename = MAKE_ARRAY(N_ELEMENTS(dateRawData), 1, /STRING, VALUE = 'NA')
  tileViewAngle = MAKE_ARRAY(N_ELEMENTS(dateRawData), 1, /FLOAT, VALUE = -9999)
  tileSunAngle = MAKE_ARRAY(N_ELEMENTS(dateRawData), 1, /FLOAT, VALUE = -9999)
  tileREdgeAnchor = MAKE_ARRAY(N_ELEMENTS(dateRawData), 1, /FLOAT, VALUE = -9999)
  tileRayleigh = MAKE_ARRAY(N_ELEMENTS(dateRawData), 1, /FLOAT, VALUE = -9999)
  tileProcessingTime = MAKE_ARRAY(N_ELEMENTS(dateRawData), 1, /STRING, VALUE = 'NA')


11. Loop through each tile within the scene. Use the j iterator to loop through each tile within the scene and print progress of the loop. 


  ; Loop through each tile, read it in, and process
  FOR j = 0, N_ELEMENTS(dateRawData) - 1 DO BEGIN


  ; Print progress of the loop
  PRINT, STRING(9B) + STRING(9B) + 'Processing steps 1 through 7 for tile' + STRCOMPRESS(STRING(j + 1) + ' of ' + STRING(N_ELEMENTS(dateRawData)))
  PRINT, STRING(9B) + STRING(9B) + STRING(9B) + STRCOMPRESS('Reading in raster as a RapidEye scene (Step 1 of 10)')











12. Determine filename and list associated data files. Based on the area of interest, RE satellite sensor, and index of uniqueDates. The data files associated with each file include an .XML file and a .JSON file. The .XML file tells ENVI/IDL how to read in the associated .TIF files for each band, after specifying the satellite platform associated with the data. The .JSON file contains the associated metadata. 


  ; Determine filename for 2_Raw_Data 
  rawDataFilename = STRCOMPRESS(aoi + '_RE_' + uniqueDates[i])
  ; List .XML files and .JSON files 
  satFile = FILE_SEARCH(FILEPATH('2_Raw_Data/' + rawDataFilename + '_rawData', ROOT_DIR = workingDir), '*metadata.xml', /FULLY_QUALIFY_PATH)
  metaFiles = FILE_SEARCH(FILEPATH('2_Raw_Data/' + rawDataFilename + '_rawData', ROOT_DIR = workingDir), '*metadata.json', /FULLY_QUALIFY_PATH)


13. Open metadata and read in acquisition time. Before reading in the .XML file, the .JSON file must be queried to determine the acquisition time of the image. 


  ; Read in metadata via .JSON file for the given tile and extract time information
  OPENR, metaFile, metaFiles[j], /GET_LUN
    metaInfo = ''
    metaLine = ''
    WHILE NOT EOF(metaFile) DO BEGIN & $
      READF, metaFile, metaLine & $
      metaInfo = [metaInfo, metaLine] & $
    ENDWHILE
  FREE_LUN, metaFile
  
  
  ; Extract the acquisition time from the metadata file
  acqTime = STRMID(metaInfo[1], STREGEX(metaInfo[1], 'acquired') + 12, 19) + 'Z'


14. Read in .XML for the given tile. Using the acquisition time queried above, read in the .XML scene by specifying the platform as ‘RapidEye.’ This is equivalent to using ENVI to open a RapidEye optical sensor from Planet Labs. Determine the filename. 


  ; Read in the raster from satFile using acqTime as the acquisition time and setting the sensor as RapidEye
  raster = e.OpenRaster(satFile[j], TIME = EnviTime(ACQUISITION = acqTime), DATASET_NAME = 'RapidEye')
  ; Determine the file basename by subsetting uniqueDates 
  fileBasename = STRCOMPRESS(aoi + '_RE_' + uniqueDates[i] + '_' + STRING(j + 1), /REMOVE_ALL)

15. Test if and how much the satellite tile intersects the extent ROI. Before processing the tile, the script first checks that the tile intersects the extent ROI read in above, and then checks how much they overlap. If they overlap by at least 20%, the raster is first cropped to the extent ROI before continuing processing. 


  ; Determine if the tile is within extentRoi
  rasterRoiExtent = extentRoi.GetExtent(raster)


  ; If raster does not fall within extentRoi, do not continue processing this file
  IF(rasterRoiExtent EQ !NULL) THEN BEGIN
  
  
    ; Print a statement that the tile will not be processed
    PRINT, STRING(9B) + STRING(9B) + 'This tile will not be processed as it does not fall within the ROI for the given estuary.'
  
  
  ; If raster does fall within the extent of estuariesRoi, continue processing.
  ENDIF ELSE BEGIN
        
        
    ; Check if the tile covers at least 20% of extentRoi before continuing processing 
    ; Determine the number of pixels in extentRoi that intersect demInput
    pixelsInROI = extentRoi.PixelCount(demInput)
    ; Find the area of these pixels assuming each DEM pixel is approximately 900 m, which is sufficient for the current task 
    areaInROI = pixelsInROI * (900 ^ 2)
    ; Determine the number of pixels in extentRoi that intersect raster
    pixelsOverlap = extentRoi.PixelCount(raster)
    ; Find the area of these pixels based on a spatial resolution of 6.5 m
    areaOverlap = pixelsOverlap * (6.5 ^ 2)
    ; Find the percent overlap between areaInROI and areaOverlap 
    pctOverlap = FLOAT(areaOverlap) / FLOAT(areaInROI)
   
        
  ; If the scene covers less than 20% of the ROI, do not continue processing 
  IF(pctOverlap LT 0.2) THEN BEGIN
        
        
    ; Print a statement that the tile will not be processed
    PRINT, STRING(9B) + STRING(9B) + 'This tile will not be processed as it covers less than 20% of the ROI for the given estuary.'

        
    ; If raster covers more than 20% of the extent of estuariesRoi, continue processing.
    ENDIF ELSE BEGIN


      ; Print a statement that the tile will be processed
      PRINT, STRING(9B) + STRING(9B) + 'This tile will be cropped to the ROI for the given estuary and processed.'
      ; Crop by extentRoi to reduce processing
      PRINT, STRING(9B) + STRING(9B) + STRING(9B) + 'Masking and cropping scene to match AOI extent (Step 2 of 10)'
      rasterCropped = ENVIROIMaskRaster(raster, extentRoi)   
          

15. Update metadata. Sun elevation and the data ignore value need to be included in the metadata. The first metadata tag specified is the 'SUN ELEVATION' tag. The code checks to see if the metadata already contains this tag. If it does, the tag is updated to the sun elevation included in the .JSON file. If it does not have this tag, it is added as the sun elevation included in the .JSON file. The 'DATA IGNORE VALUE' tag is also checked. If it already exists, it is updated to 999999.0. If it does not exist, it is added as 999999.0. 


  ; Update sun elevation metadata for rasterCropped
  IF rasterCropped.Metadata.HasTag ('SUN ELEVATION') THEN BEGIN
    rasterCropped.metadata.UpdateItem, 'SUN ELEVATION', STRMID(metaInfo[1], STREGEX(metaInfo[1], 'sun_elevation') + 15, 13)
  ENDIF ELSE BEGIN
    rasterCropped.metadata.AddItem, 'SUN ELEVATION', STRMID(metaInfo[1], STREGEX(metaInfo[1], 'sun_elevation') + 15, 13)
  ENDELSE
  ; Update data ignore value metadata for rasterCropped
  IF rasterCropped.Metadata.HasTag ('DATA IGNORE VALUE') THEN BEGIN
    rasterCropped.metadata.UpdateItem, 'DATA IGNORE VALUE', 999999.0
  ENDIF ELSE BEGIN
    rasterCropped.metadata.AddItem, 'DATA IGNORE VALUE', 999999.0
  ENDELSE







16. Perform radiometric calibration. Radiometric calibration is applied to the imagery by applying gains and offsets provided in the imagery metadata. The ENVI Task is defined as RadiometricCalibration and the input raster is specified.  The calibration type is set to Top-of-Atmosphere Reflectance and the task is executed. A new folder is created, and the results are output to this folder. 
Documentation on this task can be found at: https://www.harrisgeospatial.com/docs/ENVIRadiometricCalibrationTask.html. There is also an IF statement before applying radiometric calibration. IDL does not allow an output file to have the same name as an existing file, and there is no option to overwrite the output like in other programming languages. Therefore, these IF statements will be found any time an output will be created. If the file already exists, the code within the IF statement will not be executed. This also assists in cutting down processing time for files that already exist (i.e. if a later step needs to be run, but not the earlier steps). If a step needs to be rerun, the files or folder for that particular step and file should be deleted before running the script; otherwise, the script will check for its existence, see that it already exists, and not recreate the file even if there was a change to the script itself. 


  ; Apply radiometric calibration to convert to TOA Reflectance 
  PRINT, STRING(9B) + STRING(9B) + STRING(9B) + 'Applying radiometric calibration (Step 3 of 10)'
  IF ~ FILE_TEST(FILEPATH('3_Rad_Cal/' + fileBasename + '_radCal/' + fileBasename + '_radCal.til', ROOT_DIR = workingDir)) THEN BEGIN
    ; Call the RadiometricCalibration ENVITask
    Task_RadCal = ENVITask('RadiometricCalibration')
    Task_RadCal.INPUT_RASTER = rasterCropped
    Task_RadCal.CALIBRATION_TYPE = 'Top-of-Atmosphere Reflectance'
    Task_RadCal.Execute
    ; Output results into new folder
    radCalOutput = Task_RadCal.OUTPUT_RASTER
    FILE_MKDIR, FILEPATH('3_Rad_Cal/' + outputFile + '_radCal', ROOT_DIR = workingDir)
    radCalOutput.Export, FILEPATH('3_Rad_Cal/' + outputFile + '_radCal/' + outputFile + '_radCal.til', ROOT_DIR = workingDir), 'ENVI'
  ENDIF













17. Convert RTOA to w by dividing all values by pi. To convert from top-of-atmosphere reflectance to normalized water leaving reflectance, all cell values across all bands must be divided by pi. Unfortunately, IDL does not offer a simplified workflow to apply band math across all bands, therefore, the raster must be deconstructed into single band rasters, band math must be applied to each raster, and the bands must be stacked back together. First, the output from the previous step is read in as a raster. 

  ; Convert from R to normalized water leaving reflectance by dividing all values of each band by pi
  PRINT, STRING(9B) + STRING(9B) + STRING(9B) + 'Converting R to Rrs (Step 4 of 10)'
  IF ~ FILE_TEST(FILEPATH('4_R_To_Rrs/' + fileBasename + '_rToRrs/' + fileBasename + '_rToRrs.til', ROOT_DIR = workingDir)) THEN BEGIN
    ; Read in radiometric calibration result 
    rrsRasterFile = workingDir + '/3_Rad_Cal/' + fileBasename + '_radCal/' + fileBasename + '_radCal.til'
    rrsRaster = e.OpenRaster(rrsRasterFile)


An empty dictionary and an empty list are defined to populate results for each band. 


    ; Define an empty dictionary and an empty list to populate with  results for each band
    rrsAggregator = Dictionary()
    rrsList = List()


The ENVI Task is defined as ExtractBandsFromRaster to extract each individual band from the stacked raster. Documentation on this task can be found at: https://www.harrisgeospatial.com/docs/ENVIExtractBandsFromRasterTask.html.


    ; Extract each band from rrsRaster
    Task_ExtractBands = ENVITask('ExtractBandsFromRaster')
    Task_ExtractBands.INPUT_RASTER = rrsRaster
    Task_ExtractBands.Execute












Each band is looped through and band math is applied. The expression is to divide all pixel values in each band by pi. These results are added to the dictionary and list initiated previously. Documentation on this task can be found at: https://www.harrisgeospatial.com/docs/ENVIPixelwiseBandMathRasterTask.html.


    ; Loop through each band and apply band math
    FOREACH k, Task_ExtractBands.OUTPUT_RASTERS, k_index DO BEGIN
      ; Apply band math to divide all pixel values by pi
      Task_BandMath = ENVITask('PixelwiseBandMathRaster')
      Task_BandMath.INPUT_RASTER = k
      Task_BandMath.EXPRESSION = 'b1 / !pi'
      Task_BandMath.Execute
      ; Add results to rrsList
      rrsList.Add, Task_BandMath.OUTPUT_RASTER, /EXTRACT
      rrsAggregator.OUTPUT = rrsList
    ENDFOREACH

































The ENVI Task is defined as BuildBandStack to combine all bands into a stacked raster. However, the process of extracted and building the raster stack removes the associated metadata. Therefore, all metadata from the original rrsRaster is added to the new output from the band math stack. This raster is then output to a new folder. Documentation on this task can be found at: https://www.harrisgeospatial.com/docs/ENVIBuildBandStackTask.html.


   ; Stack each band back together using BuildBandStack
   Task_Stack = ENVITask('BuildBandStack')
   Task_Stack.INPUT_RASTERS = rrsAggregator.OUTPUT
   Task_Stack.OUTPUT_RASTER_URI = "*"
   Task_Stack.Execute
   bandMathOutput = Task_Stack.OUTPUT_RASTER
   ; Transfer metadata from rrsRaster to bandMathOutput
   FOR metadataTag = 0, N_ELEMENTS(rrsRaster.Metadata.Tags) - 1 DO BEGIN
     IF bandMathOutput.Metadata.HasTag (rrsRaster.Metadata.Tags[metadataTag]) THEN BEGIN
       bandMathOutput.Metadata.UpdateItem, rrsRaster.Metadata.Tags[metadataTag], rrsRaster.Metadata[rrsRaster.Metadata.Tags[metadataTag]]
     ENDIF ELSE BEGIN
       bandMathOutput.Metadata.AddItem, rrsRaster.Metadata.Tags[metadataTag], rrsRaster.Metadata[rrsRaster.Metadata.Tags[metadataTag]]
     ENDELSE
   END
  ; Output results into new folder
  FILE_MKDIR, FILEPATH('4_R_To_Rrs/' + outputFile + '_rToRrs/', ROOT_DIR = workingDir)
  bandMathOutput.Export, FILEPATH('4_R_To_Rrs/' + outputFile + '_rToRrs/' + outputFile + '_rToRrs.til', ROOT_DIR = workingDir), 'ENVI'
  ENDIF

















18. Find the red edge anchor value for the tile. As mentioned previously, there are three main components of dark object subtraction atmospheric correction that can be adjusted. We have already specified the Rayleigh exponent which is one component. The other two components include the wavelength at which the scattering factor is computed, and the value applied to this wavelength to represent atmospheric contamination. For this project, the red edge band is defined as the band representative of atmospheric contamination. NIR bands are also commonly used, but we have found these satellites to not be sensitive enough to collect information in these wavelengths, meaning their reflectance is often zero over dark targets. Thus, the red edge band was chosen. In order to attempt to automate the dark object subtraction workflow, the raster tile is subset both spatially and spectrally before defining the lowest 5% of the red edge distribution as the darkest pixels within the scene. First, the AOI_estuaries.XML ROI read in earlier is used to spatially subset the scene. This step generally separates water and land. 


  ; Get just water pixels by masking the raster by estuary ROI to subset spatially and masking by NDWI values to subset spectrally
  PRINT, STRING(9B) + STRING(9B) + STRING(9B) + 'Masking raster image to just water pixels (Step 5 of 10)'
  IF ~ FILE_TEST(FILEPATH('5_Mask_Raster/' + fileBasename + '_maskRaster/' + fileBasename + '_maskRaster.til', ROOT_DIR = workingDir)) THEN BEGIN
    ; Read in R to Rrs result 
    shpMaskRasterFile = FILEPATH('4_R_To_Rrs/' + outputFile + '_rToRrs/' + outputFile + '_rToRrs.til', ROOT_DIR = workingDir)
    shpMaskRasterInput = e.OpenRaster(shpMaskRasterFile)
    shpMaskRaster = ENVIROIMaskRaster(shpMaskRasterInput, estuariesRoi)
 





















The estuaries shapefile leaves some pixels along the shoreline that still contain land signal. Thus, an NDWI threshold is applied. Pixels with an NDWI value above zero are classified as water; pixels at or below zero are classified as land. The ENVIPixelwiseBandMathRaster function is used to compute NDWI as a band ratio using the NIR and Green bands. The 'MaskRaster' ENVI Task is called to mask pixels with an NDWI value at or below zero. Results are output to a new folder. Documentation on this task can be found at: https://www.harrisgeospatial.com/docs/ENVIMaskRaster.html.


    ; Calculate NDWI on shpMaskRaster to subset the scene spectrally 
    Task_NDWI = ENVITask('PixelwiseBandMathRaster')
    Task_NDWI.INPUT_RASTER = shpMaskRaster
    Task_NDWI.EXPRESSION = 'float(b2 - b5) / float(b2 + b5)'
    Task_NDWI.Execute
    rasterNDWI = Task_NDWI.OUTPUT_RASTER
    ; Create a mask raster based on an NDWI threshold
    ndwiThreshold = [0]
    ndwiThresholdRaster = ENVIBinaryLTThresholdRaster(rasterNDWI, ndwiThreshold)
    ; Apply ndwiThresholdRaster as a threshold mask on shpMaskRaster
    Task_Mask = ENVITask('MaskRaster')
    Task_Mask.DATA_IGNORE_VALUE = 999999.0
    Task_Mask.INPUT_MASK_RASTER = ndwiThresholdRaster
    Task_Mask.INPUT_RASTER = shpMaskRaster
    Task_Mask.INVERSE = 'TRUE'
    Task_Mask.Execute
    ; Output results into new folder
    maskOutput = Task_Mask.OUTPUT_RASTER
    FILE_MKDIR, FILEPATH('5_Mask_Raster/' + fileBasename + '_maskRaster/', ROOT_DIR = workingDir)
    maskOutput.Export, FILEPATH('5_Mask_Raster/' + fileBasename + '_maskRaster/' + fileBasename + '_maskRaster.til', ROOT_DIR = workingDir), 'ENVI'
  ENDIF
















Next, the atmospheric correction reference value is computed for the given tile. The masked raster created in the previous step is read in, and the red edge band data is extracted. These values are subset to just those less than 999999 (our data ignore value) and sorted. The lowest 5% of the distribution is retained and this value is reduced by a magnitude of two. This represents the red edge anchor value for the given tile. 


  ; Extract band four (red edge) values and save to list
  PRINT, STRING(9B) + STRING(9B) + STRING(9B) + 'Computing atmospheric correction reference value (Step 6 of 10)'
  ; Read in results from 5_Mask_Raster
  redEdgeFile = FILEPATH('5_Mask_Raster/' + fileBasename + '_maskRaster/' + fileBasename + '_maskRaster.til', ROOT_DIR = workingDir)
  redEdgeInput = e.OpenRaster(redEdgeFile, DATA_IGNORE_VALUE = 999999.0)
  ; Subset to red edge values and get data greater than 999999 (our data ignore value)
  redEdgeValues = redEdgeInput.GetData(BANDS = [3])
  redEdgeValuesSubset = redEdgeValues[WHERE(redEdgeValues LT 999999.0)]
  redEdgeValuesSort = redEdgeValuesSubset[SORT(redEdgeValuesSubset)]
  ; Retrieve half of the lowest 5% of the data within the red edge
  redEdgeAnchor = MEDIAN(redEdgeValuesSort[0: (N_ELEMENTS(redEdgeValuesSubset) / 20) - 1]) / 2


19. Extract and export scene information. The arrays created near the beginning of the script are populated with information about the scene. This comes from metaInfo file read in earlier in the script. After populating these arrays, the script exits the IF statement that checks if the tile covers at least 20% of the scene, the IF statement that checks if the tile covers any portion of the scene, and the FOR loop that iterates through tiles contained within the scene. The script will loop bac through the FOR loop until all tiles within the scene are processed before moving to the next step. 


        ; Add image information to their respective arrays
        tileFilename[j] = outputFile
        tileViewAngle[j] = FLOAT(STRMID(metaInfo[1], STREGEX(metaInfo[1], 'view_angle') + 13, 8))
        tileSunAngle[j] = FLOAT(STRMID(metaInfo[1], STREGEX(metaInfo[1], 'sun_elevation') + 16, 13))
        tileREdgeAnchor[j] = redEdgeAnchor
        tileRayleigh[j] = rayleighExp
        tileProcessingTime[j] = SYSTIME()

    
      ENDELSE
        
    ENDELSE
    
  ENDFOR
After looping though each tile contained within the scene, scene information is saved as a CSV file with a row per tile.


  ; Create a CSV file of the results
  FILE_MKDIR, FILEPATH('File_Information/', ROOT_DIR = workingDir)
  fileinfo_output = FILEPATH('File_Information/' + STRMID(outputFile, 0, STRLEN(outputFile) - 2) + '_fileInformation.csv', ROOT_DIR = workingDir)
  WRITE_CSV, fileinfo_output, tileFilename, tileViewAngle, tileSunAngle, tileREdgeAnchor, tileRayleigh, tileProcessingTime, $
  HEADER = ['FILENAME', 'VIEWANGLE', 'SUNANGLE', 'REDEDGEANCHOR', 'RAYLEIGH', 'PROCESSINGTIME']


20. Find the red edge anchor value for the scene. Previously, we found the red edge anchor value for the tile. Now, we will find the red edge anchor value for the scene. It is likely that not all tiles will contain an area of dark water truly representative of atmospheric contamination. Therefore, this script finds the darkest pixels possible within a tile and the minimum of these dark pixels for all tiles within the scene is used to represent the atmospheric contamination at the time of image acquisition. This is an attempt to correct for any tiles that may not contain dark water. Given that the lowest 5% of pixel values was extracted, this approach is appropriate when at least 5% of the entire scene contains dark water. If more than 95% of the entire scene is not dark water, this approach is not recommended. The wavelengths are defined for RapidEye, and the minimum of all tile red edge anchor values is found. This minimum value is used to compute the scatteringFactor based on the wavelength at the red edge band (band 3 for RapidEye) and the rayleighExp defined at the beginning of the script. The scattering factor is then applied to the wavelength at each band to determine how much reflectance is subtracted at each pixel. 


  ; Find the minimum rededge value for the entire overpass and compute the atmospheric correction values
  ; Define wavelengths 
  wavelengths = [475,555,657.5,710,805]
  ; Compute the scattering factor based on the minRededge value
  minRedEdgeAnchor = MIN(tileREdgeAnchor)
  scatteringFactor = wavelengths[3] ^ rayleighExp * minRedEdgeAnchor
  ; Use the scatteringFactor to compute the correction values for each band
  atmCorrValues = MAKE_ARRAY(N_ELEMENTS(wavelengths), 1, /FLOAT, VALUE = -9999)
  FOR j = 0, N_ELEMENTS(wavelengths) - 1 DO BEGIN
    atmCorrValues[j] = scatteringFactor / wavelengths[j] ^ rayleighExp
  ENDFOR




21. Apply atmospheric correction. Most of the processing was done on a tile-by-tile basis. However, to determine the red edge anchor value, we had to determine the red edge value for each tile before coming out of the FOR loop to determine the minimum red edge value for the entire scene. Now that we have determine the red edge anchor value for the entire scene and computed the atmospheric correction values for each band, we have to go back into a FOR loop to apply these corrections to each tile. Results from 4_R_To_Rrs are listed and looped through. 


  ; List processed files from 4_R_To_Rrs for further processing 
  rToRrs_files = FILE_SEARCH(workingDir + '\4_R_To_Rrs\', '*.TIL')


  ; Loop through each tile in rToRrs_files and apply additional processing 
  FOR j = 0, N_ELEMENTS(rToRrs_files) - 1 DO BEGIN
      
      
    ; Update the progress of the loop
    PRINT, STRING(9B) + STRING(9B) + 'Processing step 8 through 10 for tile' + STRCOMPRESS(STRING(j + 1) + ' of ' + STRING(N_ELEMENTS(satFile)))

  
    ; Define fileBasename
    fileBasename = STRMID(FILE_BASENAME(rToRrs_files[j],'.TIL'),0, STRLEN(FILE_BASENAME(rToRrs_files[j],'.TIL'))-7)


      





















The 4_R_To_Rrs output is read in as a raster and the DarkSubtractionCorrection ENVI Task is called. The values used for DOS are the atmCorrValues defined in the previous step, outside the current for-loop. Results are output to a new folder. Documentation on this task can be found at: https://www.harrisgeospatial.com/docs/ENVIDarkSubtractionCorrectionTask.html.


  ; Apply atmospheric correction
  PRINT, STRING(9B) + STRING(9B) + STRING(9B) + STRCOMPRESS('Applying atmospheric correction (Step 8 of 10)')
  IF ~ FILE_TEST(FILEPATH('6_Atm_Corr/' + fileBasename + '_atmCorr/' + fileBasename + '_atmCorr.til', ROOT_DIR = workingDir)) THEN BEGIN
    ; Read in raster results from 4_R_To_Rrs as atmospheric correction input
    atmCorrFile = rToRrs_files[j]
    atmCorrInput = e.OpenRaster(atmCorrFile[0])
    ; Apply the DarkSubtractionCorrection task
    Task_atmCorr = ENVITask('DarkSubtractionCorrection')
    Task_atmCorr.INPUT_RASTER = atmCorrInput
    Task_atmCorr.VALUES = atmCorrValues
    Task_atmCorr.Execute
    ; Output results into new folder
    atmCorrOutput = Task_atmCorr.OUTPUT_RASTER
    FILE_MKDIR, FILEPATH('6_Atm_Corr/' + outputFile + '_atmCorr/', ROOT_DIR = workingDir)
    atmCorrOutput.Export, FILEPATH('6_Atm_Corr/' + outputFile + '_atmCorr/' + outputFile + '_atmCorr.til', ROOT_DIR = workingDir), 'ENVI'
  ENDIF




















22. OPTIONAL – RPC Orthorectification. This step is set to optional because it is not required for classification and can take an extremely long time to process. Note, this step is required if mosaicing is desired. The atmospheric correction output is read in as input before calling the RPCOrthorectification ENVI Task. The output pixel size is set to the native resolution of the imagery, 2 m. The results are output to a new folder. Documentation on this task can be found at: https://www.harrisgeospatial.com/docs/RPCOrthorectification.html. To run this portion of the code, uncomment this chunk (Ctrl + ;) and re-compile. After applying RPC orthorectification, the IF statement checking the overlap between the raster and the extent ROI is closed and the FOR loop iterating through each tile in the scene.


    ; OPTIONAL -- Apply RPCOrthorectification
    PRINT, STRING(9B) + STRING(9B) + STRING(9B) + STRCOMPRESS('Applying RPC Orthorectification (Step 9 of 10)')
    IF ~ FILE_TEST(FILEPATH('7_RPC_Ortho/' + fileBasename + '_rpcOrtho/' + fileBasename + '_rpcOrtho.til', ROOT_DIR = workingDir)) THEN BEGIN
      ; Read in results from atmospheric correction
      rpcOrthoFile = FILEPATH('6_Atm_Corr/' + fileBasename + '_atmCorr/' + fileBasename + '_atmCorr.til', ROOT_DIR = workingDir)
      rpcOrthoInput = e.OpenRaster(rpcOrthoFile, DATA_IGNORE_VALUE = 999999)
      ; Apply the RPCOrthorectification task
      Task_rpcOrtho = ENVITask('RPCOrthorectification')
      Task_rpcOrtho.INPUT_RASTER = rpcOrthoInput
      Task_rpcOrtho.DEM_RASTER = demInput
      Task_rpcOrtho.OUTPUT_PIXEL_SIZE = [6.5,6.5]
      Task_rpcOrtho.Execute
      ; Output results into new folder
      rpcOrthoOutput = Task_rpcOrtho.OUTPUT_RASTER
      FILE_MKDIR, FILEPATH('7_RPC_Ortho/' + fileBasename + '_rpcOrtho/', ROOT_DIR = workingDir)
      rpcOrthoOutput.Export, FILEPATH('7_RPC_Ortho/' + fileBasename + '_rpcOrtho/' + fileBasename + '_rpcOrtho.til', ROOT_DIR = workingDir), 'ENVI'
        
    ENDIF   
           
  ENDFOR








23. OPTIONAL – Mosaic tiles into a single scene. If tiles are orthorectified, then can be mosaicked into a single raster that includes all tiles within the scene. Because this is done on a scene-by-scene basis and not a tile-by-tile basis, we exit the for-loop that was iterating through the tiles. While still inside the for-loop that is iterating through scenes, we first list any tiles from the RPC Orthorectification output for the given scene as input files for mosaicing. An empty list is initiated to populate with these rasters. This list of filenames is iterated through. Each raster is read in and added to rpcOrthoRasters. This list is then converted to an array. The BuildMosaicRaster ENVI Task is then applied to this array. Results are output to a new folder. Documentation on this task can be found at: https://www.harrisgeospatial.com/docs/ENVIBuildMosaicRasterTask.html. To run this portion of the code, uncomment this chunk (Ctrl + ;) and re-compile. After mosaicking, the FOR loop iterating through all scenes for the AOI is closed. 

  ; OPTIONAL -- Apply mosaicing
  PRINT, STRING(9B) + STRING(9B) + STRCOMPRESS('Mosaicing tiles into single scene (Step 10 of 10)')
  IF ~ FILE_TEST(FILEPATH('8_Mosaic/' + STRMID(fileBasename, 0, STRLEN(fileBasename) - 2) + '_mosaic/' + STRMID(fileBasename, 0, STRLEN(fileBasename) - 2) + '_mosaic.til', ROOT_DIR = workingDir)) THEN BEGIN
      ; List RPC Orthorectified rasters
      rpcOrthoList = FILE_SEARCH(FILEPATH('7_RPC_Ortho/' + STRMID(fileBasename, 0, STRLEN(fileBasename) - 2) + '*_rpcOrtho', ROOT_DIR = workingDir), '*rpcOrtho.til', /FULLY_QUALIFY_PATH)
      ; Create an empty list to populate with rasters
      pcOrthoRasters = List()
      ; Loop through each raster, read it in, and add to rpcOrthoRasters
      FOR k = 0, N_ELEMENTS(rpcOrthoList) - 1 DO BEGIN
        rpcOrthoRaster = e.OpenRaster(rpcOrthoList[k])
        rpcOrthoRasters.Add, rpcOrthoRaster
      ENDFOR
      ; Convert rpcOrthoRasters to an array
      mosaicInputRasters = rpcOrthoRasters.ToArray()
      ; Apply the BuildMosaicRaster task
      Task_mosaic = ENVITask('BuildMosaicRaster')
      Task_mosaic.INPUT_RASTERS = mosaicInputRasters
      Task_mosaic.Execute
      ; Output results into new folder
      mosaicOutput = Task_mosaic.OUTPUT_RASTER
      FILE_MKDIR, FILEPATH('8_Mosaic/' + STRMID(fileBasename, 0, STRLEN(fileBasename) - 2) + '_mosaic/', ROOT_DIR = workingDir)
      mosaicOutput.Export, FILEPATH('8_Mosaic/' + STRMID(fileBasename, 0, STRLEN(fileBasename) - 2) + '_mosaic/' + STRMID(fileBasename, 0, STRLEN(fileBasename) - 2) + '_mosaic.til', ROOT_DIR = workingDir), 'ENVI'
    ENDIF
  ENDFOR
24. Close ENVI and delete temporary files. Now that processing is complete, ENVI is closed. Deleting the temporary folder created at the beginning of the script has proved problematic, and this error was not resolved even with assistance from Harris Geosptial. Instead, the temporary folder can be manually deleted. Finally, the PRO called at the beginning of the script is closed.

    ; Close ENVI and delete temporary folder
    e.Close
    ;FILE_DELETE, tempFolder, /RECURSIVE
[bookmark: _GoBack]
  END

